
Abstract. This paper explores formalizing Geometric (or Clifford) alge-
bras into the Lean 3 theorem prover, building upon the substantial body
of work that is the Lean mathematics library, mathlib. As we use Lean
source code to demonstrate many of our ideas, we include a brief intro-
duction to the Lean language targeted at a reader with no prior expe-
rience with Lean or theorem provers in general. We formalize the mul-
tivectors as the quotient of the tensor algebra by a suitable relation,
which provides the ring structure automatically, then go on to establish
the universal property of the Clifford algebra. We show that this is quite
different to the approach taken by existing formalizations of Geometric
algebra in other theorem provers; most notably, our approach does not
require a choice of basis. We go on to show how operations and structure
such as involutions, versors, and the Z2-grading can be defined using the
universal property alone, and how to recover an induction principle from
the universal property suitable for proving statements about these def-
initions. We outline the steps needed to formalize the wedge product
and N-grading, and some of the gaps in mathlib that currently make
this challenging.

1. Introduction

Geometric Algebra provides a language for manipulating geometry alge-
braically, which unifies many geometric applications of complex numbers,
quaternions, and linear algebra into a single mathematical language.

There are software packages in various languages today which support
working with and developing intuition for Geometric Algebra. Most of these

packages are designed for numerical computation, typically targeted some-
where along the spectrum with ease of use at one end and efficiency at
the other. However, it is the packages designed for symbolic manipulation
which are most relevant to the topic of this paper, such as the packages for
Python [4], Maple [1], and Mathematica [2]. These packages are built on top
of computer algebra systems (CAS), and typically consist of definitions of
the objects and operators of GA, and sometimes some specialized routines to
simplify expressions.

A key feature in a CAS is the ability to simplify expressions to save
the user from doing so manually, which can be slow and error-prone. Such a
system is preloaded with a set of rules, such as x(y + z) = xy + xz which can
be used to manipulate expressions. These rules are in a sense axiomatic; a
CAS does not provide proofs of these rules, and so trust has to be placed in
both the author of the rules, and in the procedures used to apply them.

Theorem provers take quite a different approach—the base set of axioms
is very small (for example, those of Zermelo–Fraenkel set theory), and every
other “rewrite rule” must be proven from these. This approach drastically
reduces the amount of trust needed from the user—instead of having to place
trust in every single rewrite rule, their trust can be concentrated onto: the
theorem prover itself, its choice of axioms, and the fact that the definitions
used in the rewrite rule correspond to their established meaning1.

One such theorem prover is Lean [5], developed by Microsoft Research.
We choose to use this theorem prover for our formalization not because we
believe it is objectively better, but because of the rapidly growing library and
community forming around it, which we describe more in Sect. 5.

This paper will start in Sect. 4 by introducing the Lean language and
syntax, in order to provide a reader unfamiliar with either theorem provers
or Lean with the basic tools needed to understand relevant code snippets.
It will then summarize the scope of the Lean mathematics library mathlib

[19], and how we intend this work to fit within it. With these preliminar-
ies in place, it will outline in Sect. 6 the approach taken by previous for-
malizations of Geometric Algebra in other languages, and summarize their
strengths and shortcomings. At this point, we introduce our own Lean formal-
ization, contrasting its foundations to those of prior work, and demonstrating
its strengths and weaknesses by example. We conclude by outlining gaps in
mathlib’s API which, if filled, would help to tie together our formalization
with other branches of mathematics, as well as how ongoing work in mathlib

will open doors to further formalization goals.

2. Geometric Algebra

Geometric algebra (GA)2 provides a mechanism to extend a vector space with
a metric-dependent product operator, the geometric product. This product
combines and generalizes the properties of the conventional dot and cross

1e.g., A rule can state 1 × 1 = 2 if it defines × as +.
2Often used interchangeably with Clifford algebra.

product used in 3D euclidean space, while inheriting their geometric inter-
pretations; as an example, for two vectors a and b, the product decomposes
as

ab = a ⋅ b + a ∧ b

where a ⋅ b is a scalar and a ∧ b is a bivector analogous to a × b.
A geometric algebra is characterized by its metric, a quadratic form

Q(x) assigning a scalar value to the square of each vector x. It is typical3 for
our vectors to be a module over R, where the action of the quadratic form
can be characterized by its action on a set of orthogonal basis vectors ei.
Specifically, the quadratic form can be encoded a 3-tuple “signature” of natu-
ral numbers (p, q, r), indicating that p of the basis vectors satisfy Q(ei) = 1, q
satisfy Q(ei) = −1, and r satisfy Q(ei) = 0. For these real vector spaces, we use
the notation G(Rp,q,r) to denote an algebra of a given signature. Some com-
monly used geometric algebras for 3D geometry are G(R4,1,0), Conformal
Geometric Algebra (CGA); and G(R3,0,1), Plane based Geometric Algebra
(PGA). These are particularly valuable to applications in computer graph-
ics and robotics, as “blade” elements of the algebra represent 3D primitives
like lines, planes, and circles, while “rotor” elements of the algebra represent
translations, rotations, and reflections. Many other familiar algebras are iso-
morphic to a geometric algebra, like the complex numbers G(R0,1,0) and the
exterior algebra G(R0,0,n).

A key advantage to using GA for geometric problems is that it is rarely
necessary to explicitly separate coordinates when doing geometry, as would be
done for instance when constructing a rotation matrix or computing the inter-
section of two planes. In a numerical software package, internally representing
coordinates is unavoidable; but for symbolic manipulation, coordinates can
just be a distraction. For this reason, we would prefer a formalization that
avoids coordinates entirely.

3. Theorem Proving Software

Theorem proving software must do more than simply encode and verify math-
ematical truth; it must make doing so ergonomic for the user, and minimize
the inevitable friction when converting a pen and paper argument into a
machine-readable one. The typical workflow for this conversion is split into
two parts—translating the theorem statement, and then translating the proof.

The use of dependent type theory (as opposed to the set theory typically
used by mathematicians) helps with the theorem statements, as it provides
a mechanism to reject nonsensical statements like 1 = (2 < 3) (with a type
error like “2 < 3 is a Prop, expected an N”). Flexible notation is also valuable
in theorem statements, as mathematics is symbol-heavy; for instance, being
able to write ∑ instead of “sum” makes it easier to align paper with screen.

Where the software can really shine though is in the proofs, via interac-
tive and automated theorem proving. The former is about showing unresolved

3Although not assumed throughout this paper.

“goals” as the proof progresses, to help the user as they write and to guide
them what to do next. The latter is invaluable for discharging goals the user
views as trivial, such as (a+b)+c = (b+c)+a—while the user could manually
apply associativity and commutativity lemmas, automation prevents them
having to waste time thinking about this. The scale of automation can vary
from filling in the blanks (converting a proof of x < x + 1 into a proof of
2 < 3), to performing a CAS-like simplification, to applying a sophisticated
machine-learning model [18]. Since automation is typically implemented as
proof-generation, it doesn’t impact the trustworthiness of the prover—the
generated proof is subject to the same mechanical scrutiny as a hand-written
one. The line between interactive and automated is frequently blurred.

The Lean 3 language used in this paper is just one of many theorem
proving languages, and while it has all of the properties described above, it
is not unique in that it does so. Its use of dependent types is very similar to
Coq, and its heavy use of notation similar to Agda. Its automation tools are
currently less powerful than those of Isabelle/HOL, without an equivalent to
Isabelle/HOL’s powerful sledgehammer tactic [17].

4. A Quick Primer on Lean

In this section, we will give a very brief introduction to the Lean language,
in order to aid with reading the fragments of Lean code4 in the rest of this
paper. The Lean community website provides a much more in-depth set of
reference materials [12], as well as an in-browser Lean 3 environment5 [13].

We’ll start with a simple definition, showing how to define a value with
a given name and type. Here, we define the name two to refer to the natural
number (N) 2:

-- name : type := value (or "term")

def two : N := 2

Function types are declared with a → separating the types of their inputs and
outputs. Function values introduce their variable binders using a λ followed
by variable names. Here, we define a function that takes a natural number
and doubles it:

def double : N → N := λ a, two * a

For convenience, the language lets us introduce the variable and declare its
type in one place; so we could also have written this:

def double (a : N) : N := two * a

4Note that Lean source code makes heavy use of non-ASCII unicode characters, which
some PDF readers are unable to copy to the clipboard faithfully. Unfortunately, Lean
source code cannot be copied at all from the web version of this paper.
5Which may in future run a version of Lean incompatible with the code samples in this
paper. A compatible version can be ensured by using the configuration in our GitHub
repository in Sect. 9.

Function application is notated with juxtaposition, which has higher prece-
dence than most operators. Note that sometimes a $ is used in place of paren-
theses.

/- examples of function application -/

def four := double 2
def five := double 2 + 1
def six := double (2 + 1)
def six' := double $ 2 + 1

What makes Lean a language for theorem proving, and not just a regular
functional programming language, is its Prop type, which holds mathematical
statements.

def likely : Prop := ∃ x : N, x + x = 2
def unlikely : Prop := ∃ x : N, x*x = 2

Mathematical statements are themselves types (are permitted to appear to
the right of a colon), whose values constitute their proofs. We use the key-
word lemma instead of def when providing proofs, but the syntax is otherwise
identical:

lemma likely_proof : likely := 〈1, rfl〉
lemma unlikely_proof : unlikely := sorry

We can prove likely by claiming that 1 is a suitable x, and proving that 1 +

1 = 2 by definition (rfl is a proof that a = a). However, while we are free to
make the false statement unlikely, Lean does not allow us to prove it. Instead,
it grants us an escape with the sorry keyword, which is used to mark proofs
as omitted. Using sorry sets a contagious flag that marks a proof and all its
dependencies as incomplete, ensuring that the user eventually goes back to
fill in the missing proof. In this paper, we will use sorry in code examples to
indicate proofs not interesting to the reader, but present in our formalization.

Usually we do not separate our statements and proof as we did for likely.
The example below shows a slightly longer proof, using our earlier definition
of double:

lemma double_is_add_self : ∀ a, double a = a + a :=
by { intro a, apply two_mul }

This can be read as a function, that takes a number a and emits a proof
of a statement about that specific a. This is what it means for Lean to be
dependently-typed; the type of the result of a function can depend on its
input. When the inputs of a function are themselves proofs, this concept
of proofs as values leads naturally to the Howard-Curry correspondence; an
implication of the form P Ô⇒ Q is represented as a function that converts
proofs of P into proofs of Q.

The second line of this proof enters tactic mode using the by keyword.
Once within this mode, tactics like intro and apply can be used. These tactics
are often themselves Lean programs, and this mode is what makes Lean an
“interactive” theorem prover. When in this mode, compatible text editors will
show what is left to be proven after each tactic has been applied.

The logical foundation of Lean is the Calculus of Inductive Construc-
tions. The following example demonstrates a typical inductive construction;
an inductive type used to represent the logical or of two propositions:

/-- a pedagogical copy of the built-in `or` -/
inductive my_or (P Q : Prop) : Prop
| left (p : P) : my_or
| right (q : Q) : my_or

This reads as “there are two ways to construct a proof of P ∨ Q; by
providing a proof of P (my_or.left p), and by providing a proof of Q (my_or.right
q)”. From this definition, Lean provides us with an induction principle my_or.rec

which allows this procedure to be reversed, as “To prove6 P ∨Q → R, it is
sufficient to prove P → R and Q→ R”.

These “inductive” types are used to build almost every object needed
in formalized mathematics and computer science; examples include the inte-
gers, lists, logic operators, and existential quantifiers. Sometimes we need one
more fundamental building block, “quotient” types. These allow associating
an equivalence relation with a type, for instance to ensure that 1

2
= 2

4
. Lean

supports these too, and we will see more of them in Sect. 7.2.

5. Lean’s Mathematical Library mathlib

By itself, Lean is very much “batteries not included”. Its standard library is
not opinionated on whether you use it for mathematics or software verifica-
tion, and as a result comes with little beyond basic data types (e.g. int, list)
and constructive logic. Notably, it does not contain proofs of statements like
“the integers form a ring”, nor even the definition of “ring” required to make
such a statement! This is not a short-coming in Lean—it’s just a choice of
division of responsibility.

The mathematically interesting statements come from mathlib, which
is “a community-driven effort to build a unified library of mathematics for-
malized in the Lean proof assistant” [19], spanning topics like linear algebra,
topology, and analysis. The library is constantly growing [14], with over 100
contributors whose backgrounds are widely spread across both fields and
seniority, with valuable contributions coming from undergraduates and pro-
fessors alike. Perhaps a particularly notable property of mathlib is that one
of its goals is to formalize the entirety of a particular undergraduate mathe-
matics curriculum [15].

The coherency and breadth of this library makes formalization work in
Lean particularly attractive, as new structures can often be built as a thin
layer on top of existing structures. Furthermore, it reduces the effort needed
to bring formalizations in different branches of mathematics together—for
instance, in the process of formalizing statements about the wedge product
in this work (Sect. 7.8), the mathlib definition of the matrix determinant was
shown to be an alternating map.

6Here and in Lean, ∨ is the logical disjunction, not the regressive product of GA.

Table 1. Comparison of metrics between standard libraries
of popular provers using data from their GitHub repositories
as of July 2021

Language Library Source lines1 Comment lines1 Contributors2

Lean mathlib 388k 100k 159
Agda agda-stdlib 66.9k 20.4k 98
Coq math-comp 95.9k 7.6k 35

Hol-light3 707k 40.4k 8
Isabelle AFP4 4.26M 108k 386

1 Counted using the scc program.
2 According to git. Some projects do not record all authors here.
3 The standard library and language are hard to distinguish, so the metrics include both.
4 The “Archive of Formal Proofs” is not a standard library but a collection of libraries,

and as such is larger but less interlinked.

Lean is not unique in having a standard library of mathematics—two
obvious contenders are the math-comp of Coq, and the stdlib of Agda. A
thorough comparison of the three libraries is beyond the scope of this paper,
but a very rough estimate of breadth and accessibility to new users can
be obtained by counting the lines of code, comments, and contributors, as
shown in Table 1. This estimate is a poor one, as differences between the
languages themselves could easily result in the same idea taking a different
number of lines to express—but it’s sufficient to demonstrate that Lean is at
least on-par with its competitors. For reference we also include Hol-light and
Isabelle’s “Archive of Formal Proofs” (AFP) in Table 1, but as noted in the
table caption these are not suited for direct comparison.

Our intent in formalizing Geometric Algebra is for our formalization
to not only interface well with mathlib, but also for certain portions to ulti-
mately end up a part of it. Contributing code into mathlib ensures its ongoing
maintenance [20], and has numerous advantages: community review by Lean
experts, automated detection of bad practices by software tools, and gener-
ated documentation published to the web. Perhaps most important though is
that the community as a whole takes on the responsibility of keeping our con-
tributions compatible with the rest of mathlib as it evolves. The quality bar for
inclusion into mathlib is high, and as is common in software development, the
review process favors lots of small contributions over one giant contribution.
In practice this means it works best to develop larger formalizations outside
mathlib, while in parallel continually shifting parts of their foundations back
into mathlib; this allow rapid iteration unimpeded by reviewer delay, while
still insuring against the formalization diverging irrecoverably from mathlib.

In the rest of this section, we will introduce the elementary parts of
mathlib that will be essential to both translating existing formalizations and
constructing our own.

Table 2. Notation for homomorphisms in mathlib

Notation Name Properties of the map A→ B

A →l[R] B linear_map R A B Preserves +, 0, and scaling by R

A →a[R] B alg_hom R A B Preserves +, ×, 1, 0, and scaling by R

A ≃ B equiv A B Has an associated map f−1 ∶ B → A which
is a left- and right- inverse to f

A ≃l[R] B linear_equiv R A B The properties of both A →l[R] B and A ≃ B

5.1. Algebraic Structures

Some of the simplest definitions provided by mathlib are those of the alge-
braic structure hierarchy—statements like “the type M is a monoid if it is
a semi-group, has an identity 1, and ∀m ∶ M,m ⋅ 1 = 1 ⋅m = 1”. These defi-
nition are intended to be used through type-classes, which are a mechanism
for associating structure with a type. This mechanism comes in two parts: a
syntax using [] for finding type-classes,

lemma difference_of_squares {R : Type*} [ring R] (a b : R) :
(a + b) * (a - b) = a * a - b * b := sorry

and the instance keyword for registering an instance of a type-class to custom
types.

instance : ring my_type := sorry

Combined, these declarations allow our difference_of_squares lemma to be
applied to values in mytype.

Of particular interest to formalization of geometric algebra are the module

R V and algebra R A type-classes,7 which respectively describe an R-module
structure on V (roughly a vector space) and an R-algebra structure on A.
The former is of interest as it characterizes an arbitrary vector space over
which we wish to define the geometric algebra, while the latter is a minimum
requirement which our definition must meet in order to be considered useful.

5.2. Homomorphisms

In mathematics, it is frequently useful to talk about structure-preserving
morphisms, such as an R-linear map; a function f(x) ∶ A → B that
f(x + y) = f(x) + f(y) and ∀c ∈ R,f(cx) = cf(x). In mathlib this is rep-
resented by the type linear_map R A B that “bundles” the map itself with proofs
of the aforementioned properties. Since such bundled maps are quite com-
mon in mathlib, shorthand notation is provided. Table 2 summarizes mathlib’s
bundled maps and their notations used in this paper.

7Which themselves require at least [semiring R] [add_comm_monoid V] and [comm_semiring R]

[semiring A], respectively.

6. Existing Formalizations of GA

While there has been no published use of Lean to formalize geometric,
Clifford, or Grassman–Cayley algebras, there are already formalizations in
other theorem proving languages. To aid the reader, when describing these
other formalizations this section presents code snippets translated into their
(roughly) equivalent Lean code.

Of particular interest in these existing formalizations is the underly-
ing definition used to define a multivector, and how this definition can be
used to define a product of some kind. While proofs are obviously important
to a formalization, every proof has to start with a theorem statement, the
expressivity of which is limited by the definitions at hand.

We present them roughly in order of generality.

6.1. Fixed-Dimension Representations

In [11], a formalization in Isabelle/HOL specific to G(R3) is presented that
enumerates all the r-vectors explicitly by assigning a name to each grade:

@[ext]
structure multivector :=
(scalar : R) (vector : R × R × R) (bivec : R × R × R) (trivec : R)

From here, the operations of an abelian group can be defined componentwise:

instance : has_zero multivector := 〈
{ scalar := 0, vector := 0, bivec := 0, trivec := 0}〉
instance : has_add multivector := 〈λ x y,
{ scalar := x.scalar + y.scalar, vector := x.vector + y.vector,

bivec := x.bivec + y.bivec, trivec := x.trivec + y.trivec}〉
-- and for `has_sub`, `has_neg`

and shown to satisfy its axioms using the ext lemma which states “multivectors
are equal if their four components are equal”:

instance : add_comm_group multivector :=
{ zero := 0, add := (+), sub := has_sub.sub, neg := has_neg.neg,

add_zero := λ _, ext _ _ (add_zero _) (add_zero _) (add_zero _) (add_zero _),
..sorry /- the 4 other axioms are proved the same way -/ }

Already we can see that this approach inevitably leads to a lot of repetition,
as while it would be easy to generalize over the integer grades 0, 1, 2, 3, it is
challenging to generalize over the names scalar, vector, bivec, and trivec.

The problem only becomes worse when defining the geometric product,
as now we have four terms in each of the components. As in [11], the full
definition is omitted below.

instance : has_one multivector := 〈
{ scalar := 1, vector := 0, bivec := 0, trivec := 0}〉
instance : has_mul multivector := 〈λ x y,
{ scalar := x.scalar * y.scalar + dot_product x.vector y.vector

- dot_product x.bivec y.bivec - x.trivec * y.trivec,
vector := sorry, bivec := sorry, trivec := sorry}

From here, [11] goes on to prove that multivectors form a ring, by showing
that multiplication associates and distributes, and operates as expected with

one. Again, this would have to be done component-wise, and the trivial but
presumably verbose proofs are omitted from [11].

The tedium of the component-wise definitions and proofs can be reduced
by generation from CAS implementations and automation features within the
theorem provers, but this scales poorly to more complex statements, which
may need to be tackled by hand. Needless to say, this formalization does not
scale at all to other dimensions and signature of algebra, as no part of it is
generalized.

6.2. Recursive Tree Representations

A convenient way to escape this death-by-cases is to use a recursive defini-
tion of a multivector. This is the approach taken by the Coq formalization of
Grassmann–Cayley algebra in [7] (and resembles the computational approach
used by Garamon [3]). There, the definition is built as a balanced binary tree,
where each branch indicates the presence or absence of a basis blade, and the
leafs contain the corresponding coefficient.

def Gn : N → Type
| 0 := R -- a scalar coefficient
| (n + 1) := Gn n × Gn n -- the parts without and with $en$

In this Lean translation, the | syntax is used to pattern match against an
integer representing the remaining depth of the tree, while × captures the
branching. For instance, a + a1e1 + a2e2 + a12e12 is represented as a term of
type Gn 2 as ((a, a2), (a1, a12)).

The recursive data definition leads naturally to a recursive operator def-
inition, which resembles the following:8

reserve infix ` ⋏ `:70
def wedge : Π {n}, Gn n → Gn n → Gn n
| 0 x y := (x * y : R)
| (n + 1) 〈x1, x2〉 〈y1, y2〉 := let infix ` ⋏ ` := wedge in

(x2 ⋏ y2, x1 ⋏ y2 + x2 ⋏ y1
d)

infix ` ⋏ ` := wedge

Here, the pattern-matching is against the depth of the tree, and for non-
root elements, the two branches. This elegantly avoids having to deal with
coefficient-wise proofs, and results in a definition that works on algebras of
arbitrary dimension n.

While [7] does not extend to defining a metric or the geometric product
it infers, their follow-up work [8] does so. This design still shares a short-
coming of the previous design—it imbues the definition with a canonical and
orthogonal basis, which is at odds with our goal of being coordinate-free.

8The y1
d notation is introduced in [7], but not essential to get a feel for how the recursive

definitions look.

6.3. Indexed Coordinate Representations

The HOL light formalization in [16] offers the same generality as the one in
Sect. 6.2, but slightly more abstractly describing a multivector as a set of
coefficients indexed by the IDs of its basis vectors.

variables (n : N)

-- mapping from subsets of 1:n to coefficients
abbreviation idx := set (fin n)
def multivector : Type := idx n → R

[16] goes on to define the geometric product and various derived products
for arbitrary metrics, which means their formalizations can be used for both
CGA and PGA. A rough translation of their generalized product formaliza-
tion is as follows.

def generic_prod (a b : multivector n) (sgn : idx n → idx n → R) : multivector n :=
Σ ai bi in finset.univ,

pi.single (ai ∆ bi) ((a ai * b bi) * sgn ai bi)

Here ai ∆ bi is the symmetric difference. This is then used to derive the wedge
and other products as

def wedge (a b : multivector n) : multivector n :=
prod a b $ λ ai bi,

if ai ∩ bi ≠ ∅ then 0 else -- matching blades
(-1) ^ (finset.card $ (ai.product bi).filter $ λ abj, abj.1 > abj.2)

Overall, the formalization that follows in [16] is expansive, covering top-
ics ranging from the existence of inverses to outermorphisms. However, the
initial definition of multivector ingrains a preferred choice of orthogonal (finite)
basis, which while in line with many numerical packages for Clifford alge-
bras, is at odds with how vector spaces are formalized in mathlib. mathlib’s
approach is typically axiomatic, introducing explicit sets of basis vectors only
when needed; often, only the proof of the existence of a set of basis vectors
is used.

To mesh well with mathlib, our formalization will need to support alge-
bras over a variety of vector spaces, not just those with coefficients in R

n. We
are of course free to take on extra assumptions should we need them (such
as the scalars forming a field, or the dimension of the space being finite),
but by making the initial definitions more general we leave the door open
to future researchers interested in other algebras which do not satisfy these
assumptions.

7. Our Formalization of GA

In this section, we present a variety of definitions and theorems for GA that
we formalized in Lean. For reasons described in Sect. 5, some of these formal-
izations have been integrated into mathlib, while the rest are available in our
source code repository (linked in Sect. 9). We won’t attempt to distinguish
where the line is drawn, as ultimately the result is the same; users of our
source repository will automatically get a compatible version of mathlib.

7.1. Remarks on Type Theory

While it is typical on paper to avoid distinguishing “the multivectors of grade
zero” from “the scalars”, or “the multivectors of grade 1” from “the vectors”, the
strict dependent typing of Lean forces us to treat these separately. Instead of
saying that the multivectors G(V ) “contain” the scalars R and vectors V , we
need to provide explicit mappings between these distinct types. Respectively,
these come in the form of a ring homomorphism algebra_map ∶ R → G(V )
and a linear map ι ∶ V → G(V )9.

Typically we would assume these mappings are injective, but we can
obtain many results without needing to do so. As it turns out, the defini-
tions outlined below permit these maps to be non-injective, for particularly
pernicious choices of R and V [9]10.

7.2. The Quotient Definition

A Clifford algebra over the vector space V with quadratic form Q ∶ V → R

can be defined as a quotient of the tensor algebra T (V ) by the two-sided ideal
IQ generated from the set {v ⊗ v −Q(v) ∣ v ∈ V }. Generating an ideal from
this set amounts to taking the smallest superset of it that is closed under
left- and right- multiplication and addition by elements of T (V ). When we
take a quotient by an ideal, we are saying that two elements are considered
equivalent if their subtraction is in that ideal. It follows then that we have
v ⊗ v ≈ Q(v), and therefore our construction ensures that vectors square to
scalars.

As of writing, mathlib does not have direct support for two-sided ideals,
but it does support the equivalent operation of taking the quotient by a suit-
able closure of a relation like v ⊗ v ≈ Q(v). As such, the quotient definition
still translates fairly naturally into Lean:

variables {R : Type*} [comm_ring R]
variables {V : Type*} [add_comm_group V] [module R V]
variables (Q : quadratic_form R V)

inductive rel : tensor_algebra R V → tensor_algebra R V → Prop
| of (v : V) : rel

(tensor_algebra.ι R v * tensor_algebra.ι R v)
(algebra_map R (tensor_algebra R V) (Q v))

/-- `clifford_algebra Q' is the algebra over `V` with metric `Q` -/

@[derive [ring, algebra R]]
def clifford_algebra := ring_quot (clifford_algebra.rel Q)

/-- `ι Q v' is the embedding of the vector `v : V' into the Clifford
algebra with quadratic form `Q`. -/

def clifford_algebra.ι : V →l[R] clifford_algebra Q :=
(ring_quot.mk_alg_hom R _).to_linear_map.comp (tensor_algebra.ι R)

What makes this definition particularly attractive is that thanks to the
@[derive ...] attribute (a Lean metaprogram built into mathlib), the ring and
algebra structure can be automatically proved by Lean using its knowledge of
ring_quot. Note that it is this ring (clifford_algebra Q) structure which provides
the geometric product *!

9The names of which are taken from mathlib conventions.
10A partial formalization of this result was shown in Lean.

The operations described in Sect. 7.1 also fall out with minimal
effort: the derived algebra R (clifford_algebra Q) structure provides the map from
the scalars algebra_map R _ r; while the map from the vectors clifford_algebra.ι

Q v is obtained by first mapping the vectors into the tensor algebra (using
tensor_algebra.ι R, the analogous ι for the tensor algebra), and then embed-
ding them within the quotient using the API around ring_quot. Putting these
together, we can verify that our construction does indeed square vectors to
scalars:

theorem ι_sq_scalar (v : V) : ι Q v * ι Q v = algebra_map R _ (Q v) :=
begin

erw [←alg_hom.map_mul, ring_quot.mk_alg_hom_rel R (rel.of v), alg_hom.commutes],
refl,

end

The remaining code to explain is our rel definition, which demonstrates
how inductive types can be use for propositions, as was introduced briefly in
Sect. 4. Here, we declare rel T1 T2 as a proposition over pairs of elements
in T (V ), but provide only one way to construct it, rel.of v. In essence, this
means that a proof of rel T1 T2 is a proof that there exists some v such that
T1 = ι(v)⊗ι(v) and T2 =Q(v). Note that mathlib uses * not ⊗ for the product
in the tensor algebra.

While not written by the authors, a better understanding of the power
of inductive types can be obtained by looking inside mathlib’s definition of
ring_quot. The inductive type used by this definition to extend our rel defini-
tion to its closure under ring operations is roughly as follows:

inductive crel (rel : R → R → Prop) : R → R → Prop
| of {∣x y : R∣} (h : rel x y) : crel x y
| add_left {∣a b c∣} : crel a b → crel (a + c) (b + c)
| mul_left {∣a b c∣} : crel a b → crel (a * c) (b * c)
| mul_right {∣a b c∣} : crel b c → crel (a * b) (a * c)

This reads as “A pair of elements are satisfied by the closure of a relation if:”

of. “They satisfy the relation rel”
add_left. “They can each be split into an addition with the same right

operand c, and with left operands satisfying the closure of the relation.”
mul_left, mul_right. “They can each be split into an multiplication with the

same left/right operand, and with a right/left operand satisfying the
closure of the relation.”

One might remark that something still seems odd about this relation, as
neither this crel nor our rel imply reflexivity, symmetry, or transitivity —
however, these properties follow from the axiomatization of Lean’s quot type,
and are provided as part of the Lean prelude as quot.exact : quot.mk r a = quot.mk

r b → eqv_gen r a b.

7.3. The Universal Property

While convenient to define, the quotient can be hard to state further defini-
tions and prove theorems about. When defining operations over a quotient,
the approach is almost always to operate on the data within the quotient,

and then prove that for any operands that are considered equal under the
quotient, the output of the operation is unchanged.

Such proofs can be very challenging, especially given some short-comings
in Lean when it comes to recursing over nested inductive types (such as crel

which wraps our rel above). Another approach is to work with the universal
property from category theory. The universal property of the Clifford algebra
is shown diagrammatically in Eq. (7.1)—given a linear map f from V to an
algebra A satisfying f(v)f(v) = Q(v) ⋅ 1A, there is a unique addition-and-
multiplication-preserving map from G(Q) to A that satisfies f = (lift f) ○ ι.
Since lift provides a unique mapping, it has an inverse and we can also state
this property in reverse in terms of g.

G(Q) A

V

liftf=g

f=lift−1 g
ι (7.1)

The universal property can be stated in Lean as

def lift :
{f : V →l[R] A // ∀ v, f v * f v = algebra_map _ _ (Q v)}
≃ (clifford_algebra Q →a[R] A) := sorry

which with the help of Table 2, reads piecewise as:

lift : _ ≃ _ “lift” is an equivalence between...
V →l[R] A ... R-linear maps from the vector space V to the algebra A...
{f : _ // ∀ v, f v * f v = algebra_map _ _ (Q v)} ... whose output squares to the

metric ...
clifford_algebra Q →a[R] A ... and maps between G(Q) and A which are R-

linear and preserve multiplication.

The implementation of lift (as opposed to just its type) is not included in
this paper, but consists largely of invoking uninteresting machinery around
ring_quot already in mathlib. This machinery within mathlib encapsulates the
task of working under the quotient, so that we don’t have to

Note that if we apply this equivalence in reverse to the identity algebra
automorphism G(Q) → G(Q) (that is, evaluate (lift Q).symm (alg_hom.id R _)),
then we recover a linear map from V to G(Q) whose results square to scalars.
Intuitively11, this is ι(v).

The type of lift alone is enough for us to prove statements like (lift f)(a+
b ∗ c) = f(a) + f(b)f(c), while its composition with ι gives us the remaining
interesting properties.

For the rest of our formalization, our proofs and definitions depend only
on the properties of lift and ι, and never on the quotient construction from
Sect. 7.2. In principle, this would enable future work to transfer proofs of
theorems about our construction to any other construction, such as those in
Sect. 6, provided that those definitions are equipped with their own lift and
ι.

11But thankfully, also proven by Lean!

7.4. Conjugations

Beyond the identity mapping (lift Q) (ι Q) = alg_hom.id R _, the next simplest
operation we can use it to define is the grade involution X̂ , which flips the
sign of every component vector. In Lean, we write that

def involute : clifford_algebra Q →a[R] clifford_algebra Q :=
clifford_algebra.lift Q 〈-(ι Q), λ v, by simp〉

where by simp is producing the trivial proof that
-(ι Q v)*-(ι Q v) = (ι Q v)*(ι Q v). Just as we saw earlier with lift, the type of
involute alone is enough to prove statements about involutions of addition
and multiplication. This time, we also get from the type the fact that invo-
lute acts as the identity on scalars. All that remains is to prove that involution
negates vectors, which get almost for free using some @[simp] lemmas about
the universal property (not shown):

@[simp] lemma involute_ι (v : V) : involute (ι Q v) = -ι Q v :=
by simp [involute]

To check that the Lean simplifier is suitably trained for involute, we show that
taking the involution of a product of n vectors ((l.map $ ι Q).prod) is equivalent
to scaling by (−1)n:

lemma involute_prod_map_ι : ∀ l : list M,
involute (l.map $ ι Q).prod = ((-1 : R)^l.length) • (l.map $ ι Q).prod

| [] := by simp
| (x :: xs) := by simp [pow_add, involute_prod_map_ι xs]

A similar approach can be performed to define grade reversion
X̃, although this time instead of inserting a minus sign for each
vector, we need to flip the multiplication order. mathlib provides us
an op : X → Xop mapping for exactly that, which by definition satisfies
op x * op y = op (y * x). Applying this to each of our vectors will then
give op(reverse(x)), which we can map back into X with unop : X

op
→ X.

It can be trivially shown that this mapping is linear and invertible
(opposite.op_linear_equiv : X ≃l[R] Xop), which combined with some rather ugly
boilerplate gives us a complete definition for grade reversal.

def reverse : clifford_algebra Q →l[R] clifford_algebra Q :=
(op_linear_equiv R).symm.to_linear_map.comp (

clifford_algebra.lift Q 〈(opposite.op_linear_equiv R).to_linear_map.comp (ι Q),
λ m, unop_injective $ by simp〉).to_linear_map

Unlike involute, reverse is only a →l[R] and not a →a[R], so we need to prove how
it acts on scalars and multiplication ourselves. Once again, simp makes short
work of this.

@[simp] lemma reverse.map_mul (a b : clifford_algebra Q) :
reverse (a * b) = reverse b * reverse a :=

by simp [reverse]

@[simp] lemma reverse.commutes (r : R) :
reverse (algebra_map R (clifford_algebra Q) r) = algebra_map R _ r :=

by simp [reverse]

While the above definitions of involute and reverse give us proofs about
their operations on sums and products essentially for free, they miss one
key property of these conjugations; the fact that they are involutive, reverse

(reverse x) = x. Proving these becomes much easier once we build an inductive
principle.

7.5. Induction

The induction principle we seek can be stated “If a property P holds for the
algebra_map of r : R and the ι of v : V into clifford_algebra Q, and is preserved under
addition and multiplication, then it holds for all of clifford_algebra Q”. Perhaps
surprisingly, the universal property alone is enough for us to construct this
principle. An a outline of the approach is:

1. Show that collectively, the inputs to our inductive principle define a
subalgebra (s) of precisely the elements that satisfy P; that is, a subset
of the full Clifford algebra which contains zero and one and is closed
under addition and multiplication. By doing this, Lean provides us with
a type ↥s which bundles each element of the subalgebra with a proof
that it belongs to that subalgebra.

2. Restrict the codomain of ι Q to ↥s, and show that doing so still preserves
the fact that vectors square to scalars.

3. Lift this restricted map from V → ↥s to clifford_algebra Q → ↥s.
4. Apply this lifted map to our input a. We show that the value part of this

lifting is just a, meaning the proof part is C a, our proof for an arbitrary
element.

The full Lean implementation is shown below.

@[elab_as_eliminator]
lemma induction {P : clifford_algebra Q → Prop}

(h_grade0 : ∀ r, P (algebra_map R (clifford_algebra Q) r))
(h_grade1 : ∀ x, P (ι Q x))
(h_mul : ∀ a b, P a → P b → P (a * b))
(h_add : ∀ a b, P a → P b → P (a + b))
(a : clifford_algebra Q) :
P a :=

begin
-- the arguments are enough to construct a subalgebra, and a mapping into it from M

let s : subalgebra R (clifford_algebra Q) :=
{ carrier := P, mul_mem' := h_mul, add_mem' := h_add, algebra_map_mem' := h_grade0 },
let of : clifford_hom Q ↥s :=
〈 (ι Q).cod_restrict s.to_submodule h_grade1,

λ m, subtype.eq $ ι_sq_scalar Q m 〉,
-- the mapping through the subalgebra is the identity
have of_id : alg_hom.id R (clifford_algebra Q) = s.val.comp (lift Q of),
{ ext,
simp [of], },

-- finding a proof is finding an element of the subalgebra

convert subtype.prop (lift Q of a),
exact alg_hom.congr_fun of_id a,

end

Armed with our new hammer, we find a lot of nail-like lemmas we can eas-
ily prove. Below, we show that involution and reverse commute, and each is

involutive; that is, ˜̂x = ˆ̃x, ˆ̂x = x, and ˜̃x = x.

@[simp] lemma reverse_involute_commute :
function.commute (reverse : _ → clifford_algebra Q) involute :=

λ x, by induction x using clifford_algebra.induction; simp [*]

@[simp] lemma involute_involutive :
function.involutive (involute : _ → clifford_algebra Q) :=

λ x, by induction x using clifford_algebra.induction; simp [*]

@[simp] lemma reverse_involutive :
function.involutive (reverse : _ → clifford_algebra Q) :=

λ x, by induction x using clifford_algebra.induction; simp [*]

7.6. Z2-grading

An algebra A is said to be graded by an additive monoid I if there exists
a family of I-indexed submodules Ai such that x ∈ Ai, y ∈ Aj → xy ∈ Ai+j ,
1 ∈ A0, and every element a ∈ A has a unique decomposition12 a = ∑i ai
where ai ∈ Ai. If we adjust this definition to use a wedge product instead of a
regular product, then we can obtain the familiar N-grading used in Geometric
Algebra. However, we haven’t formalized the wedge product yet!

Instead, we can show that the Clifford algebra permits a Z2 grading,
that is a grading by the integers modulo two. Intuitively, this maps to the
notion of even- and odd-graded r-vectors. Again, we get there via the univer-
sal property; this time, mapping into add_monoid_algebra (clifford_algebra Q) (zmod

2) where add_monoid_algebra is built into mathlib. This type can be thought of as
somewhat analogous to a polynomial, where the coefficients are the even and
odd multivectors themselves, and the variable is either x = x3 = ⋯ (odd) or
1 = x2 = ⋯ (even), where multiplication behaves as addition does in Z2. We
write this in Lean as follows, where the name single indicates constructing an
object of a single grade.

def grades' :
(clifford_algebra Q) →a[R] add_monoid_algebra (clifford_algebra Q) (zmod 2) :=

(lift Q : _) 〈
-- vectors are grade 1
(add_monoid_algebra.lsingle 1).comp (ι Q),
-- this requires 1 + 1 = 0, which is why we use `zmod 2` as our grading
λ x, by { simp [add_monoid_algebra.single_mul_single], congr }〉

Thanks to the universal property, and the fact it produces an →a[R] for us, we
get obvious properties like grades' (r + s) i = grades' r i + grades' s i for free.

To finish this definition of the grading, we show that there is →a[R] in
the other direction forming a left inverse (by simply summing components),
and we show that the grades are disjoint. We state this latter fact as

@[simp]
lemma grades'.map_grades' (x : clifford_algebra Q) (i : zmod 2) :

let xi := (grades' x i) in grades' xi = finsupp.single i xi := sorry

which reads “If you take the grade i part of x (xi), and then split xi into its
constituent grades, it will consistent of a single grade i”. The Lean proof is

12Which corresponds to requiring that the submodules Ai span the space, and they satisfy
some notion of disjointness.



   28 Page 18 of 26 Eric Wieser and Utensil Song Adv. Appl. Clifford Algebras

omitted in this paper for brevity, but it once again falls out after application
of the induction principle from Sect. 7.5.

7.7. Versors

The versors can be described as a set of multivectors closed under multipli-
cation (submonoid) and scaling, generated from the set of vectors (set.range (ι

Q)). We write this formally as:

def versors := center_submonoid.closure (set.range (ι Q))

where center_submonoid is a wrapper for a set of elements, built on top of the
submonoid and sub_mul_action structures in mathlib that respectively carry proofs
of closure under multiplication and scaling.

Lean provides some useful syntax for working with sub-objects like this.
Instead of working with (x : clifford_algebra Q) (hx : x ∈ versors Q), we can write
as a shorthand (v : ↥versors Q). The advantage of “bundling” the multivector
with its proof of being a versor like this is that operations v * w and k • v can
be defined to automatically produce a bundle containing a proof that their
result is also a versor; and indeed, center_submonoid does just that.

Armed with our definition, the next step is to once again con-
struct an induction principle. Most of the heavy lifting is done by the
submonoid.closure_induction' principle in mathlib, which was developed as part
of this paper. The statement of this principle is:

/-- If a statement `C` is true for all scalars, all vectors, and all products
of versors which each satisfy `C`, then it is true for all versors. -/
lemma induction_on {C : versors Q → Prop} (v : versors Q)

(h_scalars : ∀ r : R, C 〈algebra_map _ _ r, (versors Q).algebra_map_mem r〉)
(h_vectors : ∀ m, C 〈ι Q m, ι_mem m〉)
(h_mul : ∀ a b, C a → C b → C (a * b)) :
C v := sorry

The induction principle lets us once again bash out some useful state-
ments with uninteresting proofs. This time, we scratch the surface of Lean’s
meta-programming framework to avoid repeating a trivial proof:

/-- A simple macro tactic that we can reuse between proofs -/

meta def inv_rev_tac : tactic unit :=
`[apply induction_on v,

{ intro r, simp, },
{ intro m, simp, },
{ intros a b ha hb, simp [(versors Q).mul_mem, ha, hb] }]

/-- Involute of a versor is a versor -/

@[simp] lemma involute_mem (v : versors Q) :
involute (v : clifford_algebra Q) ∈ versors Q := by inv_rev_tac

/-- Reverse of a versor is a versor -/
@[simp] lemma reverse_mem (v : versors Q) :

reverse (v : clifford_algebra Q) ∈ versors Q := by inv_rev_tac

A more interesting application of our induction principle is to prove that
the product of a versor and its reverse is a scalar:

/-- A versor times its reverse is a scalar -/

lemma mul_self_reverse (v : versors Q) :
∃ r : R, (v : clifford_algebra Q) * reverse (v : clifford_algebra Q) = ↑ar :=

begin
with_cases { apply induction_on v },
case h_grade0 : r {
refine 〈r * r, _〉,
simp },

case h_grade1 : m {
refine 〈Q m, _〉,
simp },

case h_mul : x y {
rintros 〈(qx : R), hx〉 (〈qy : R), hy〉, -- results for `x` and `y` by induction

refine 〈qx * qy, _〉,
simp only [reverse_mul, submonoid.coe_mul, ring_hom.map_mul],
rw [mul_assoc ↑x, ←mul_assoc ↑y, hy, algebra.commutes, ←mul_assoc, hx], }

end

Here we use some more verbose Lean syntax to clearly indicate each branch
of the induction. From here, we go on to show that versors have an inverse,
and that for a non-trivial algebra over a field in an anistropic metric, they
form a group with zero (i.e. all elements but zero have an inverse):

instance
{K} [field K] {V} [add_comm_group V] [module K V]
{Q : quadratic_form K V} [nontrivial (clifford_algebra Q)] [f : fact Q.anisotropic] :
group_with_zero (versors Q) := sorry

7.8. Wedge Products and N-Grading

Formalizing the wedge product of n vectors can be done as follows,

/-- A wedge product of n vectors. Note this does not define the wedge product of

arbitrary multivectors. -/↪

def ι_wedge (n : N) [invertible (n.factorial : R)] :
alternating_map R M (clifford_algebra Q) (fin n) :=

1/(n! : R) •

((mk_pi_algebra_fin R n _).comp_linear_map (λ i, ι Q)).alternatization

The definitions and API for alternating maps, and crucially their construction
from the alternatization of a multilinear map (multilinear_map.alternatization)
were some of the many contributions to mathlib as part of the work in this
paper. Through this API, we can easily prove statements like
∀ a b, ι_wedge _ ![a, b, a] = 0 (i.e. a ∧ b ∧ a = 0 for all vectors a, b), or more
generally that the wedge product of linearly dependent vectors is zero.

Of course, a crucial use of the wedge product is between arbitrary
multivectors, not just basis vectors. The ideal way to define this wedge
product would be to find a linear isomorphism to_ext : clifford_algebra Q ≃l[R]

exterior_algebra R V, and implement wedge by transferring the arguments along
the isomorphism, computing the product in the exterior algebra, and trans-
ferring back:

def wedge (a b : clifford_algebra Q) : clifford_algebra Q :=
to_ext.symm (a.to_ext * b.to_ext)

The advantage of this approach is that theorems about multiplication in exte-
rior algebras can be transferred easily to theorems about the wedge product
in Clifford algebras.

Our work does not go as far as defining the to_ext, although in col-
laboration with members of the Lean community the authors contributed a
definition of exterior_algebra R V to mathlib using a similar approach to that
described in Sects. 7.2 and 7.3. Constructing the isomorphism to_ext from the
universal property alone is difficult.

One way to do so would be to first show that the Clifford algebra is a
filtered algebra, in that it can be split into subsets X0 ⊆ X1 ⊆ X2 ⊆ ⋯ such
that a ∈ Xi, b ∈ Xj Ô⇒ ab ∈ Xi+j . From here, the associated graded algebra

can be constructed with grades Y0 =X0, Yn+1 =
Xn+1

Xn
where division indicates

the quotient submodule. Finally, a variant of the Poincaré–Birkhoff–Witt
theorem can be applied to find the isomorphism.

The definition of a filtered algebra, along with a proof that a Clifford
algebra satisfies it with Xi as “the multivectors of at most grade i” was
formalized in Lean by the authors. The associated graded algebra proved
challenging to encode in dependent type theory, as the type of each grade
becomes a function of the type of the grade before it. This obstacle is not
insurmountable, but the authors feel it best left to future work.

Interestingly, it can be shown that such an isomorphism does not always
exist; as mentioned in Sect. 7.1, there exist Clifford algebras under our def-
inition where the “algebra map” is not injective. However, we show in our
formalization that the “algebra map” of the exterior algebra is always injec-
tive. Since only one of the two maps is always injective, these pathological
Clifford algebras cannot be isomorphic to the corresponding exterior algebra.

7.9. Constructing Specific Algebras

In Sect. 2, we draw attention to some particularly useful geometric algebras
over the real numbers, which until this point our formalization has been too
general to mention. In this section, we show how to use our general formal-
ization with specific algebras. It is worth noting that our formalization does
not permit us to make statements like “the complex numbers are a geomet-
ric algebra”, instead requiring us to rephrase to “the complex numbers are
isomorphic to a geometric algebra”. We state this isomorphism in Lean as
follows:

/-- The quadratic form sending elements to the negation of their square. -/
noncomputable def complex.Q : quadratic_form R R :=
-quadratic_form.lin_mul_lin linear_map.id linear_map.id

@[simp] lemma complex.Q_apply (r : R) : complex.Q r = -(r*r) := rfl

/-- The clifford algebra with quadratic form `complex.Q` is isomorphic to `C`. -/

def equiv_complex : clifford_algebra complex.Q ≃a[R] C := sorry

lemma equiv_complex_ι : complex (ι complex.Q 1) = complex.I := sorry

The full details are provided in our source repository, as well as a similar
construction for the quaternions. This choice to work with isomorphisms is
pragmatic one; it’s consistent with how mathlib handles the tensor algebra,
and it doesn’t force us to declare one particular isomorphism as canonical; in
the example above, we could just as easily mapped ι complex.Q 1 to -complex.I

instead of complex.I.

For a more illustrative example than the complex numbers, we will use
the rest of this section to outline how to set up a definition of Conformal
Geometric Algebra, G(Rn+1,1,0), which augments the initial vector space R

n

with two extra dimensions spanned by the orthogonal vectors e+ and e− where
e2+ = 1 and e2− = −1. In practice it can be algebraically convenient to span these
extra dimensions instead with null basis vectors no, n∞ where no ⋅n∞ = 1, and
n2
o = n

2
∞ = 0, as is done in [6, Table 13.1]. This approach is desirable because

these basis vectors have more geometric meaning, with no being associated
with the origin, and n∞ associated with the point at infinity.

We start by defining the conformalized vector space of a real vector
space as the triple of (original vector space V , n0 coefficient, n∞ coefficient).
Note that by doing this we are choosing a preferred basis (something we gen-
erally wanted to avoid) over the extra dimensions, but we continue to avoid
doing so over V .

@[derive [add_comm_group, vector_space R]]
def conformalize (V : Type*) [inner_product_space R V] : Type* := V × R × R

We proceed by providing linear maps to extract each component:

def v : conformalize V →l[R] V := linear_map.fst _ _ _
def c_n0 : conformalize V →l[R] R :=

(linear_map.fst _ _ _).comp (linear_map.snd _ _ _)
def c_ni : conformalize V →l[R] R :=

(linear_map.snd _ _ _).comp (linear_map.snd _ _ _)

and some definitions to construct conformal vectors:

/-- The embedding of direction vectors into `conformalize V`. -/

def of_v : V →l[R] conformalize V := linear_map.inl _ _ _
/-- The no basis vector. -/
def n0 : conformalize V := (0, 1, 0)
/-- The n∞ basis vector. -/
def ni : conformalize V := (0, 0, 1)

Finally, we can define the up mapping up(v) = n0 + v +
1

2
∥v∥

2
n∞ and the

conformal metric Q, the final pieces needed to construct the Clifford algebra:

/-- The embedding of points from `V` to `conformalize V`. -/
def up (x : V) : conformalize V :=
n0 + of_v x + (1 / 2 * ∥x∥^2 : R) • ni

/-- The metric is the metric of V plus an extra term about n0 and ni. -/

def Q : quadratic_form R (conformalize V) :=
(bilin_form_of_real_inner.comp v v).to_quadratic_form

- (2 : R) • quadratic_form.lin_mul_lin c_n0 c_ni

variables (V)
/-- Define the Conformal Geometric Algebra over `V` . -/
abbreviation CGA := clifford_algebra (Q : quadratic_form R (conformalize V))

With our definitions out of the way, our next job is to train the Lean
simplifier about trivial combinations of these functions

@[simp] lemma v_of_v (x : V) : (of_v x).v = x := rfl
@[simp] lemma c_n0_of_v (x : V) : (of_v x).c_n0 = 0 := rfl
@[simp] lemma c_ni_of_v (x : V) : (of_v x).c_ni = 0 := rfl
-- 6 more lemmas follow combining {v, c_n0, c_ni} with {n0, ni}

From this, we can prove that Q has the form we’d expect, that up correctly
produces null vectors, and that the metric between two conformal points is
proportional to their distance:

@[simp] lemma Q_apply (x : conformalize V) : Q x = ∥x.v∥^2 - 2 * (x.c_n0 * x.c_ni) :=
by simp [Q, inner_self_eq_norm_sq_to_K]

@[simp] lemma Q_up (x : V) : Q (up x) = 0 :=
by simp [up, ←mul_assoc]

lemma Q_polar_up (x y : V) : quadratic_form.polar Q (up x) (up y) = -dist x y ^ 2 :=
sorry

Note that while at a glance this last result appears off by a factor of two
from the “usual” result that up(x) ⋅up(y) = − 1

2
∥x − y∥, this is because Lean’s

quadratic_form.polar Q x y is defined as twice the value of the inner product x ⋅y.

A construction of PGA in Lean can be done in very much the same way,
and is included in our source repository.

8. Future Work

In Sect. 7, we formalized a variety of elementary GA operations, and showed
how to leverage various existing definitions supplied in mathlib to do so in
a way that is concise and integrates with the rest of the library. Our work
however, is not yet at the point to be ready for use in formalizing results
at the edge of research in Clifford Algebras, as some important elementary
definitions are still absent. In this section, we will outline the absent areas of
formalization which would contribute significantly to closing this gap.

8.1. Graded Modules and Algebras

To date, mathlib does not have a reusable definition for graded modules and
algebras, making it difficult to make definitions that can be reused across
tensor, exterior, and Clifford algebras. In Sect. 7.6, we present a definition
that represents the Z2-gradation using a injective map into add_monoid_algebra

(clifford_algebra Q) (zmod 2). This representation is flawed, as there are terms of
this type that do not represent a valid grading result such as single 1 r for
some scalar r, which falsely claims that the scalars are a grade one element.
This can be partially resolved by restricting this injective map to its range
to form an equivalence, but the resulting object becomes awkward to work
with.

A better solution would be to use mathlib’s direct_sum type in place of
add_monoid_algebra, as this allows us to put a different restriction upon each
grade. Unfortunately, unlike the latter, the former is not equipped with a
multiplicative structure needed to use it with the universal property. Design-
ing the API for equipping such a structure was deemed out of scope for
this paper, but will be essential for conveniently working with the grades of
geometric, exterior, and tensor algebras in Lean.

8.2. Isomorphisms from Clifford to Exterior Algebras

As mentioned in Sect. 7.8, this work was not able to define the wedge product
on multivectors, largely due to desiring a coordinate-free approach. While this
makes our formalization less useful than the other formalizations in Sect. 6,
it should be noted that those formalizations require an explicit coordinate
parametrization. To make progress in this direction, mathlib needs to acquire
definitions and API for working with filtered algebras and their associated
graded algebra. Linking up the definitions in mathlib of the Clifford alge-
bra with the definition of the universal enveloping Lie algebra would help
with applying the Poincaré–Birkhoff–Witt theorem, which itself would need
formalizing.

8.3. Geometric Calculus

Parallel to our efforts to formalize geometric algebra, mathlib has been mak-
ing great progress in the field of analysis, with slowly improving support
for multivariate calculus. A notable result in the last few months was the
formalization of variations of the Fundamental Theorem of Calculus.

As this part of mathlib matures, it should be possible to formalize the
Geometric Calculus described by [10]. This will however not be a simple
task—many of the results are stated in terms of an is_R_or_C typeclass, which
in practice means they apply only to the reals or the complex numbers. The
next step towards formalizing Geometric Calculus would be to generalize
some of these results to the quaternions (which are in mathlib), which would
inform how to sensibly break apart the is_R_or_C structure into smaller pieces.

9. Conclusions

We have shown how Lean can be used to formalize Geometric Algebra in a
variety of ways. Of these ways, we chose to develop the formalization using
a quotient definition, as this offered greater generality than any previous
formalization. We demonstrate how to use the universal property of the Clif-
ford algebra that this definition provides to recover additional definitions and
operators, and how to prove that these constructions behave as we intend. To
demonstrate the generality of our approach, we show how to specialize it to
a Conformal Geometric Algebra over a vector space of arbitrary dimension.
Throughout, we make liberal use of the mathlib library for Lean, demonstrat-
ing the convenience of having an expansive body of formalized mathematical
terminology stated coherently in a single place.

A complete archive consisting of our formalization, our translations of
parts of existing formalizations, the precise versions of Lean and mathlib that
our code is compatible with, and a summary of the various contributions to
mathlib made as part of this work can be found on GitHub at

https://github.com/pygae/lean-ga

mathlib is not only growing quickly, but changes quickly too. Through having
contributed the key definitions from our work back into mathlib, we ensured

https://github.com/pygae/lean-ga

these definitions will change with it, and can be built upon in future by
others.

